Binomialverteilung

DieBinomialverteilung ist ein typisches Beispiel einer diskreten Wahrscheinlichkeitsverteilung.

(%i1) kill(all)$

Problembeschreibung

Für gegebene Werte von n und p sind
a) die Dichtefunktion der Binomialverteilung
b) die Binomialverteilung
c) der Erwartungswert
d) die Varianz und
e) die Streuung der Binomialverteilung
zu bestimmen.

Problemlösung

UNTERPROGRAMM LADEN

(%i1) load(distrib);

(%o1)

C:/Programme/Maxima-5.17.1/share/maxima/5.17.1/share/contrib/distrib/distrib.mac

Unterprogramme ermöglichen die Nutzung zusätzlicher Funktionen, fördern jedoch die Gefahr des „BLACK BOX“-Denkens

EINGABE

(%i2) n:read("Gib den Umfang der Verteilung ein");

Gib den Umfang der Verteilung ein 10;

(%o2) 10

(%i3) p:read("Gib die a-priori-Wahrscheinlichkeit ein");

Gib die a-priori-Wahrscheinlichkeit ein 1/6;

(%o3) $\frac{1}{6}$

Das könnte z.B. bedeuten: „Es wird 10mal gewürfelt“. Jede Binomialverteilung ist durch die Parameter n und p definiert.

VERARBEITUNG

(%i4) Dichte(x):=pdf_binomial(x,n,p);

(%o4) Dichte(x):= pdf_binomial(x , n , p)

Druchsichtiger ist: **Dichte(x):=binomial(n,k)*p**k*(1-p)**(n-k)**

(%i5) Verteilung(x):=cdf_binomial(x,n,p);

(%o5) Verteilung(x):= cdf_binomial(x , n , p)

Diese Funktionen sind im Unterprgramm DISTRIB enthalten.

(%i6) D:makelist(Dichte(x),x,0,n);

(%o6) $\left[\frac{9765625}{60466176}, \frac{9765625}{30233088}, \frac{1953125}{6718464}, \frac{390625}{2519424}, \frac{546875}{10077696}, \frac{21875}{1679616}, \frac{21875}{10077696}, \frac{625}{2519424}, \frac{125}{6718464}, \frac{25}{30233088}, \frac{1}{60466176}\right]$

(%i7) D:D,numer;

(%o7) [0.16150558288985 , 0.32301116577969 , 0.29071004920172 , 0.15504535957425 , 0.054265875850988 , 0.013023810204237 , 0.0021706350340395 , $2.4807257531880302\ 10^{-4}$, $1.8605443148910226\ 10^{-5}$, $8.2690858439601012\ 10^{-7}$, $1.6538171687920201\ 10^{-8}$]

(%i8) V:makelist(Verteilung(x),x,0,n);

(%o8) [cdf_binomial$\left(0, 10, \frac{1}{6}\right)$, cdf_binomial$\left(1, 10, \frac{1}{6}\right)$, cdf_binomial$\left(2, 10, \frac{1}{6}\right)$, cdf_binomial$\left(3, 10, \frac{1}{6}\right)$, cdf_binomial$\left(4, 10, \frac{1}{6}\right)$, cdf_binomial$\left(5, 10, \frac{1}{6}\right)$, cdf_binomial$\left(6, 10, \frac{1}{6}\right)$, cdf_binomial$\left(7, 10, \frac{1}{6}\right)$, cdf_binomial$\left(8, 10, \frac{1}{6}\right)$, cdf_binomial$\left(9, 10, \frac{1}{6}\right)$, 1]

Numerische Auswertung schafft Klarheit:

(%i9) V:V,numer;

(%o9) [0.16150558288985 , 0.48451674866954 , 0.77522679787126 , 0.93027215744551 , 0.9845380332965 , 0.99756184350074 , 0.99973247853478 , 0.99998055111009 , 0.99999915655324 , 0.99999998346183 , 1]

(%i10) Erwartungswert:mean_binomial(n,p);

(%o10) $\frac{5}{3}$

Die Berechnung kann auch herkömmlich stattfinden: E = n.p

(%i11) Erwartungswert:Erwartungswert,numer;

(%o11) 1.666666666666667

Wir runden das Ergebnis noch:

(%i12) Erwartungswert:floor(Erwartungswert*1000+0.5)/1000.0;

(%o12) 1.667

(%i13) Varianz:var_binomial(n,p);

(%o13) $\frac{25}{18}$

Die herkömmliche Berechnung wäre V = n.p.(1-p)

(%i14) Varianz:Varianz,numer;

(%o14) 1.388888888888889

(%i15) Varianz:floor(Varianz*1000+0.5)/1000.0;

(%o15) 1.389

(%i16) Streuung:std_binomial(n,p);

(%o16) $\frac{5}{3\sqrt{2}}$

Die Streuung ist die Wurzel aus der Varianz!

(%i17) Streuung:Streuung,numer;

(%o17) 1.178511301977579

(%i18) Streuung:floor(Streuung*1000+0.5)/1000.0;

(%o18) 1.179

Rundung auf drei Stellen nach dem Komma

AUSGABE

(%i19) print("Dichtefunktion =",D)$

Dichtefunktion = [0.16150558288985 , 0.32301116577969 , 0.29071004920172 , 0.15504535957425 , 0.054265875850988 , 0.013023810204237 , 0.0021706350340395 , $2.4807257531880302\ 10^{-4}$, $1.8605443148910226\ 10^{-5}$, $8.2690858439601012\ 10^{-7}$, $1.6538171687920201\ 10^{-8}$]

(%i20) print("Verteilung =",V)$

Verteilung = [0.16150558288985 , 0.48451674866954 , 0.77522679787126 , 0.93027215744551 , 0.9845380332965 , 0.99756184350074 , 0.99973247853478 , 0.99998055111009 , 0.99999915655324 , 0.99999998346183 , 1]

(%i21) print("Erwartungswert =",Erwartungswert)$

Erwartungswert = 1.667

(%i22) print("Varianz =",Varianz)$

Varianz = 1.389

(%i23) print("Streuung =",Streuung)$

Streuung = 1.179

Binomialverteilung mit Grafik

(%i24) kill(all)$

Problembeschreibung

Gegeben sind n und p einer Binomialverteilung.
Berechne die Binomialverteilung und stelle diese grafisch dar.

Problemlösung

EINGABE

(%i1) n:read("Gib den Parameter n ein");

Gib den Parameter n ein 10;

(%o1) 10

(%i2) p:read("Gib die Wahrscheinlichkeit p ein");

Gib die Wahrscheinlichkeit p ein 1/6;

(%o2) $\frac{1}{6}$

VERARBEITUNG

(%i3) W(k):=binomial(n,k)*p**k*(1-p)**(n-k);

(%o3) $W(k):=\binom{n}{k} p^{k} (1-p)^{n-k}$

Das ist die bekannte Formel für die Binomialverteilung.

(%i4) kumW(k):=sum(W(i),i,0,k);

(%o4) $\mathrm{kumW}(k):=\sum_{i=0}^{k} W(i)$

AUSGABE

(%i5) for j:0 thru n do
print(j,?truncate(W(j)*1000+0.5)/1000.0,?truncate(kumW(j)*1000+0.5)/1000.0);

0 0.162 0.162

1 0.323 0.485

2 0.291 0.775

3 0.155 0.93

4 0.054 0.985

5 0.013 0.998

6 0.002 1.0

7 0.0 1.0

8 0.0 1.0

9 0.0 1.0

10 0.0 1.0

(%o5) *done*

?truncate() sollte eher durch floor() ersetzt werden

(%i6) wxplot2d([W(x)], [x,0,n])$

(%t6)



Die grafische Darstellung mit wxMaxima ist im Text eingebettet. Die Darstellung unterstellt eine stetige Funktion, in Wirklichkeit handelt es sich um eine diskrete Wahrscheinlichkeitsverteilung.

Binomialverteilung

(%i7) kill(all)$

Problembeschreibung

Von einer Binomialverteliung kennt man die Parameter n und p.
Zu erstellen sind:
a) die Verteilung
b) die kumulierte Verteilung
c) eine Grafik der Verteilung
d) ein **definierter Teilbereich der Verteilung**

Problemlösung

EINGABE

(%i1) n:read("Gib den Parameter n ein");

Gib den Parameter n ein 10;

(%o1) 10

(%i2) p:read("Gib die Wahrscheinlichkeit p ein");

Gib die Wahrscheinlichkeit p ein 1/6;

(%o2) $\frac{1}{6}$

VERARBEITUNG

(%i3) W(k):=binomial(n,k)*p**k*(1-p)**(n-k);

(%o3) $W(k) := \binom{n}{k} p^{k} (1-p)^{n-k}$

(%i4) kumW(k):=sum(W(i),i,0,k);

(%o4) $\mathrm{kumW}(k) := \sum_{i=0}^{k} W(i)$

AUSGABE

(%i5) for j:0 thru n do
print(j,floor(W(j)*1000+0.5)/1000.0,floor(kumW(j)*1000+0.5)/1000.0);

```
0 0.162 0.162
1 0.323 0.485
2 0.291 0.775
3 0.155 0.93
4 0.054 0.985
5 0.013 0.998
6 0.002 1.0
7 0.0 1.0
8 0.0 1.0
9 0.0 1.0
10 0.0 1.0
```

(%o5) *done*

Die Wahrscheinlichkeitsverteilung kann grafisch dargestellt werden:

(%i6) wxplot2d([W(x)], [x,0,n])$

(%t6)



Die Darstellung der Dichtefunktion. Die stetige Darstellung darf nicht darüber hinwegtäuschen, dass es sich um eine diskrete Verteilungsfunktion handelt.

(%i7) wxplot2d([kumW(x)], [x,0,n])$

(%t7)



Die Darstellung der kumulierten Verteilung.

Häufig benötigt man nur Teilbereiche der gesamten Verteilung:

EINGABE

```
(%i8) a:read("Gib die Untergrenze ein");
 Gib die Untergrenze ein 0;
(%o8) 0

(%i9) b:read("Gib die Obergrenze ein");
 Gib die Obergrenze ein 5;
(%o9) 5
```

AUSGABE

```
(%i10) for j:a thru b do
   print(j,floor(W(j)*1000+0.5)/1000.0,floor(kumW(j)*1000+0.5)/1000.0);
0 0.162 0.162
1 0.323 0.485
2 0.291 0.775
3 0.155 0.93
4 0.054 0.985
5 0.013 0.998
(%o10) done
```

(%i11) summe:sum(W(k),k,a,b),numer;

(%o11) 0.99756184350074

Diese Technik wird bei vielen Aufgabestellungen benötigt.

(%i12) summe:floor(summe*1000+0.5)/1000.0;

(%o12) 0.998

(%i13) print("Summe von",a,"bis",b,"=",summe)$

Summe von 0 *bis* 5 = 0.998

Binomialverteilung

(%i13) kill(all)$

Problembeschreibung

Von einer Binomialverteilung kennt man n und p,
sowie die Untergrenze a und die Obergrenze b.

Die kumulierte Wahrscheinlichkeit über diesen Bereich
ist gesucht.

Problemlösung

EINGABE

(%i1) n:read("Gib den Parameter n ein");

Gib den Parameter n ein 10;

(%o1) 10

(%i2) p:read("Gib die Wahrscheinlichkeit p ein");

Gib die Wahrscheinlichkeit p ein 1/6;

(%o2) $\frac{1}{6}$

(%i3) a:read("Gib die Untergrenze ein");

Gib die Untergrenze ein 1;

(%o3) 1

(%i4) b:read("Gib die Obergrenze ein");

Gib die Obergrenze ein 4;

(%o4) 4

VERARBEITUNG

(%i5) k:makelist(i,i,a,b);

(%o5) [1 , 2 , 3 , 4]

(%i6) P(k):=binomial(n,k)*p**k*(1-p)**(n-k);

(%o6) $P(k):=\binom{n}{k} p^{k} (1-p)^{n-k}$

(%i7) p:makelist(P(i),i,a,b);

(%o7) $\left[\frac{9765625}{30233088}, \frac{1953125}{6718464}, \frac{390625}{2519424}, \frac{546875}{10077696}\right]$

(%i8) n:length(p);

(%o8) 4

(%i9) s:sum(p[i],i,1,n);

(%o9) $\frac{49765625}{60466176}$

(%i10) ausgabe:[transpose(k),transpose(p)]$

Transpose() tauscht Zeilendarstellung gegen Spaltendarstellung.

AUSGABE

(%i11) print(ausgabe)$

$$\left[\begin{bmatrix}1\\2\\3\\4\end{bmatrix},\begin{bmatrix}\frac{9765625}{30233088}\\ \frac{1953125}{6718464}\\ \frac{390625}{2519424}\\ \frac{546875}{10077696}\end{bmatrix}\right]$$

(%i12) print("P(",a,"<= k <=",b,") =",s)$

$$P(\ 1\ <=k<=\ 4\ )=\frac{49765625}{60466176}$$

Binomialverteilung: 3-mal würfeln

(%i13) kill(all)$

Problembeschreibung

Wir würfeln 3-mal. Wie groß ist die Wahrscheinlichkeit, dass
a) 0 Sechser kommen,
b) 1 Sechser kommt,
c) 2 Sechser kommen oder
d) 3 Sechser kommen?

Problemlösung

BERECHNUNG MIT BERNOULLIKETTE

(%i1) W:{0,1};

(%o1) $\{0,1\}$

(%i2) P[1]:1/6;

(%o2) $\frac{1}{6}$

(%i3) P[0]:1-P[1];

(%o3) $\frac{5}{6}$

(%i4) S:cartesian_product(W,W,W);

(%o4) {[0,0,0],[0,0,1],[0,1,0],[0,1,1],[1,0,0],[1,0,1],[1,1,0],[1,1,1]}

Die Produktmenge ist CARTESIAN_PRODUCT(). S ist das sichere Ereignis. 0 bedeutet, es kommt **nicht Kopf**, 1 bedeutet, es **kommt Kopf**.

(%i5) A(k):=subset(S,lambda([e],is(e[1]+e[2]+e[3]=k)));

(%o5) $A(k) := \text{subset}\left(S, \text{lambda}\left([e], \text{is}\left(e_1 + e_2 + e_3 = k\right)\right)\right)$

Es müsste eigentlich auch A(k):=subset(S,k) funktionieren.

(%i6) A:makelist(A(k),k,0,3);

(%o6) [{[0,0,0]},{[0,0,1],[0,1,0],[1,0,0]},{[0,1,1],[1,0,1],[1,1,0]},{[1,1,1]}]

(%i7) B:transpose(A);

(%o7) $\begin{bmatrix} \{[0,0,0]\} \\ \{[0,0,1],[0,1,0],[1,0,0]\} \\ \{[0,1,1],[1,0,1],[1,1,0]\} \\ \{[1,1,1]\} \end{bmatrix}$

(%i8) B[1];

(%o8) [{[0,0,0]}]

(%i9) P[0]*P[0]*P[0];

(%o9) $\frac{125}{216}$

(%i10) B[2];

(%o10) [{[0,0,1],[0,1,0],[1,0,0]}]

(%i11) P[0]*P[0]*P[1]+P[0]*P[1]*P[0]+P[1]*P[0]*P[0];

(%o11) $\frac{25}{72}$

(%i12) B[3];

(%o12) [{[0,1,1],[1,0,1],[1,1,0]}]

(%i13) P[0]*P[1]*P[1]+P[1]*P[0]*P[1]+P[1]*P[1]*P[0];

(%o13) $\frac{5}{72}$

(%i14) B[4];

(%o14) [{[1,1,1]}]

(%i15) P[1]*P[1]*P[1];

(%o15) $\frac{1}{216}$

BERECHNUNG MIT BINOMIALVERTEILUNGSFUNKTION

(%i16) n:3;

(%o16) 3

(%i17) p:P[1];

(%o17) $\frac{1}{6}$

(%i18) W(k):=binomial(n,k)*p**k*(1-p)**(n-k);

(%o18) $W(k):=\binom{n}{k} p^{k} (1-p)^{n-k}$

(%i19) A:makelist(W(k),k,0,n);

(%o19) $[\frac{125}{216}, \frac{25}{72}, \frac{5}{72}, \frac{1}{216}]$

(%i20) B:transpose(A);

(%o20) $\begin{bmatrix} \frac{125}{216} \\ \frac{25}{72} \\ \frac{5}{72} \\ \frac{1}{216} \end{bmatrix}$

.